

How to Query in SPARQL an Ontoterminology built with Tedi and exported in RDF



Technical Report: TR-Tedi-RDF-2020-11-v3

<http://ontoterminology.com/>

Date: ~~14/11/2020~~ ~~24/12/2020~~ 26/12/2020

Last version: http://www.ontologia.fr/OTB/SPARQL_Queries.pdf


Christophe Roche - <http://christophe-roche.fr/>

Maria Papadopoulou - <http://o4dh.com/maria-papadopoulou>

Table des matières

1. Export RDF Graph	2
2. SPARQL Endpoint.....	4
3. Protégé	5
4. Competency Questions in SPARQL.....	6
CQ1: List of English terms and definitions?	6
CQ2: List of English terms and definitions using OTV vocabulary?	7
CQ3: List of English terms and own differences of the corresponding concepts?.....	8
CQ4: List of English terms, status, definitions and names of the corresponding concepts?	9
CQ5: List of concept names and terms?.....	10
CQ6: List of concepts, definitions, terms and status?	11
CQ7: List of English terms with their status and definition using only OTV vocabulary?	12
CQ8: What is (are) the difference(s) between a 'bell krater' and a 'type B amphora'?.....	13
CQ9: What is (are) the difference(s) between a 'type B amphora' and a 'bell krater'?.....	13
CQ10: What are all the differences between a 'bell krater' and a 'type B amphora'?	14
CQ11: What are the common characteristics of bell kraters and type B amphorae?	14
CQ12: List of the first 10 black-figure vases with their images?	15
5. Export OntoLex-Lemon.....	17
1) Return the list of English terms with their definition using OntoLex-Lemon:	17

How to Query in SPARQL an Ontotermology



2. SPARQL Endpoint

<http://demo.openlinksw.com/sparql>



The screenshot shows the OpenLink Virtuoso SPARQL Query Editor interface. At the top, there is a header with the OpenLink Software logo and the text "OpenLink Virtuoso SPARQL Query Editor". Below the header, there is a navigation bar with links: "Not logged in. Login", "About", "Namespace Prefixes", "Inference Rules", "Macros", "RDF Views", "Permalink", and "iSPARQL".

The main area is divided into two sections. The top section is labeled "Default Graph URI" and contains a text input field with the value "http://www.ontologia.fr/OTB/krater.rdf" and a "Run Query" button. The bottom section is labeled "Query Text" and contains a text area with the query: "SELECT DISTINCT * WHERE {[] a ?EntityType} LIMIT 50".

Below the query text area, there is a box containing the text "FROM <http://www.ontologia.fr/OTB/krater.rdf>".

At the bottom of the interface, there are several settings: "Sponging:" with a dropdown menu set to "Retrieve all missing remote RDF data that might be useful, including seeAlso references"; "Results Format:" with a dropdown menu set to "HTML" and a note "(The CXML output is disabled, see details)"; "Execution timeout:" with a text input field set to "30000000" and a note "milliseconds (values less than 1000 are ignored)"; and "Options:" with a checkbox labeled "Strict checking of void variables".

3. Protégé



4. Competency Questions in SPARQL

CQ1: List of English terms and definitions?

<http://demo.openlinksw.com/sparql>

PREFIX rdf: <<http://www.w3.org/1999/02/22-rdf-syntax-ns#>>

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>


SELECT ?term ?definition

FROM <<http://www.ontologia.fr/OTB/krater.rdf>>

```
WHERE { ?concept rdf:type owl:Class.
        ?concept skos:prefLabel ?term.
        ?concept skos:definition ?definition.
        FILTER (lang(?term) = 'en')
        FILTER (lang(?definition) = 'en')
}
```

ORDER BY ?term

LIMIT 10



term	definition
"amphora"@en	"Vase with neck and two handles for storing and transport liquids and solids."@en
"bell krater"@en	"Krater without neck, with upward curling handles placed high on the body."@en
"belly amphora"@en	"Amphora with foot, without a clearly defined neck."@en
"calyx krater"@en	"Krater without neck, with upward curling handles placed low on the body."@en
"column krater"@en	"Krater with a clearly defined neck and column-like handles."@en

CQ2: List of English terms and definitions using OTV vocabulary?

<http://demo.openlinksw.com/sparql>

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
```

```
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
```

```
SELECT ?termName ?definition
```

```
FROM <http://www.ontologia.fr/OTB/krater.rdf>
```

```
WHERE { ?concept rdf:type otv:Concept.  
        ?concept otv:denotedByTerm ?term.  
        ?term otv:language ?lg.  
        ?term otv:termName ?termName.  
        ?term otv:termDefinition ?definition  
        FILTER (?lg = "en")  
}
```

```
ORDER BY ?term
```

```
LIMIT 10
```

termName	definition
"amphora"	"Vase with neck and two handles for storing and transport liquids and solids."
"bell krater"	"Krater without neck, with upward curling handles placed high on the body."
"belly amphora"	"Amphora with foot, without a clearly defined neck."
"calyx krater"	"Krater without neck, with upward curling handles placed low on the body."
"column krater"	"Krater with a clearly defined neck and column-like handles."

CQ3: List of English terms and own differences of the corresponding concepts?

<http://demo.openlinksw.com/sparql>

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
```

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
```

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
```

```
SELECT ?name ?diffName
```

```
FROM <http://www.ontologia.fr/OTB/krater.rdf>
```

```
WHERE {
```

```
    ?concept rdf:type otv:Concept.
```

```
    ?concept skos:prefLabel ?name.
```

```
    ?concept otv:ownDifference ?dif.
```

```
    ?dif rdfs:label ?diffName.
```

```
    FILTER (lang(?name)="en")
```

```
}
```

```
ORDER BY ?name
```

name	diffName
"amphora"@en	"with handles"
"amphora"@en	"for storing and transport"
"amphora"@en	"with neck"
"amphora"@en	"with two handles"

amphora

Definition: Vase with neck and two handles for storing and transport liquids and solids.

Concept: <Vessel for storing and transport with neck with handle(s) with two handles>

essential characteristic(s): /vessel/, /for storing and transport/, /with neck/, /with handle(s)/, /with two handles/,

a kind of: <Vessel>,

linked to: <Place>, <Handle>, <Neck>,

CQ4: List of English terms, status, definitions and names of the corresponding concepts?

<http://demo.openlinksw.com/sparql>

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
```

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
```

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
```

```
SELECT DISTINCT ?lg ?termLabel ?status ?def ?cptName
```

```
FROM <http://www.ontologia.fr/OTB/krater2.rdf>
```

```
WHERE {
  ?concept rdf:type otv:Concept.
  ?concept otv:conceptName ?cptName.
  ?concept otv:denotedByTerm ?term.
  ?term otv:term ?termLabel.
  ?term otv:termDefinition ?def.
  ?term otv:language ?lg.
  ?term otv:termStatus ?status.
  FILTER (?lg="en")
  FILTER (?status="preferred")
}
```

```
ORDER BY ?termLabel
```

lg	termLabel	status	def
en	"amphora"	preferred	"Vase with neck and two handles for storing and transport liqu
en	"bell krater"	preferred	"Krater without neck, with upward curling handles placed high
en	"belly amphora"	preferred	"Amphora with foot, without a clearly defined neck."
en	"calyx krater"	preferred	"Krater without neck, with upward curling handles placed low o
en	"column krater"	preferred	"Krater with a clearly defined neck and column-like handles."

cptName
"<Vessel for storing and transport with neck with handle(s) with two handles>"
"<Vessel for mixing wine with water with handle(s) with foot with open mouth without neck with upward curling
"<Vessel for storing and transport with neck with handle(s) with two handles without a clearly defined neck>"
"<Vessel for mixing wine with water with handle(s) with foot with open mouth without neck with upward curling
"<Vessel for mixing wine with water with handle(s) with foot with open mouth with neck with a clearly defined

CQ5: List of concept names and terms?

Protégé

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

PREFIX owl: <http://www.w3.org/2002/07/owl#>

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

PREFIX otv: <http://www.ontologia.fr/OTB/otv#>

SELECT ?conceptName ?termLabel

WHERE { ?subject rdf:type otv:Concept .

?subject otv:denotedByTerm ?term.

?term rdfs:label ?termLabel.

?subject otv:conceptName ?conceptName.}

ORDER BY ?conceptName

SPARQL query:

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
```

```
SELECT ?conceptName ?termLabel
WHERE {
  ?subject rdf:type otv:Concept .
  ?subject otv:denotedByTerm ?term.
  ?term rdfs:label ?termLabel.
  ?subject otv:conceptName ?conceptName.}
```

ORDER BY ?conceptName

conceptName	termLabel
'<Person>'	"personne"@fr
'<Place>'	"lieu"@fr
'<Vessel for mixing wine with water with handles with foot with open mouth with neck with a clearly defined neck>'	"cratère à colonnettes"@fr
'<Vessel for mixing wine with water with handles with foot with open mouth with neck with a clearly defined neck>'	"κρατήρας κιονωτός"@gr
'<Vessel for mixing wine with water with handles with foot with open mouth with neck with a clearly defined neck>'	"column krater"@en
'<Vessel for mixing wine with water with handles with foot with open mouth with neck with a clearly defined neck>'	"κρατήρας ελικοειδής"@gr
'<Vessel for mixing wine with water with handles with foot with open mouth with neck with a clearly defined neck>'	"volute krater"@en
'<Vessel for mixing wine with water with handles with foot with open mouth with neck with a clearly defined neck>'	"cratère à volutes"@fr
'<Vessel for mixing wine with water with handles with foot with open mouth without neck with upward curling han>'	"cratère en cloche"@fr
'<Vessel for mixing wine with water with handles with foot with open mouth without neck with upward curling han>'	"κρατήρας κωδωνόσχημος"@gr
'<Vessel for mixing wine with water with handles with foot with open mouth without neck with upward curling han>'	"bell krater"@en
'<Vessel for mixing wine with water with handles with foot with open mouth without neck with upward curling han>'	"calyx krater"@en
'<Vessel for mixing wine with water with handles with foot with open mouth without neck with upward curling han>'	"cratère en calice"@fr
'<Vessel for mixing wine with water with handles with foot with open mouth without neck with upward curling han>'	"κρατήρας καλυκοειδής"@gr
'<Vessel for mixing wine with water with handles with foot with open mouth>'	"krater"@en
'<Vessel for mixing wine with water with handles with foot with open mouth>'	"κρατήρας"@gr
'<Vessel for mixing wine with water with handles with foot with open mouth>'	"cratère"@fr
'<Vessel for perfume with neck>'	"λήκυθος"@gr
'<Vessel for perfume with neck>'	"lekythos"@en
'<Vessel for perfume with neck>'	"lécythos"@fr

CQ6: List of concepts, definitions, terms and status?

Protégé

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
```

```
SELECT ?subject ?def ?label ?status
WHERE {
  ?subject rdf:type otv:Concept.
  ?subject otv:denotedByTerm ?term.
  ?term rdfs:label ?label.
  ?term otv:termStatus ?status.
  ?term skos:definition ?def
  FILTER (lang(?label)='en')}
```

```
ORDER BY ?subject
```

subject	def	label	status
Vessel	"Ceramic vessel."@en	"vase"@en	"preferred"
Vessel_for_mixing_wine_with_water_with_handle(s)_with_1	"Large open container with a foot and handles, used in an "	"krater"@en	"preferred"
Vessel_for_mixing_wine_with_water_with_handle(s)_with_1	"Krater with a clearly defined neck and column-like handles."	"column krater"@en	"preferred"
Vessel_for_mixing_wine_with_water_with_handle(s)_with_1	"Krater with a clearly defined neck and volute-like handles."	"volute krater"@en	"preferred"
Vessel_for_mixing_wine_with_water_with_handle(s)_with_1	"Krater without neck, with upward curling handles placed h "	"bell krater"@en	"preferred"
Vessel_for_mixing_wine_with_water_with_handle(s)_with_1	"Krater without neck, with upward curling handles placed lc "	"calyx krater"@en	"preferred"
Vessel_for_perfume_with_neck	"Vase with neck for perfume."@en	"lekythos"@en	"preferred"
Vessel_for_prize_with_foot_with_handle(s)_with_two_hanc	"Vase with foot, two handles, and neck, which served as pi "	"panathenaic amphora"@en	"alternative"
Vessel_for_prize_with_foot_with_handle(s)_with_two_hanc	"Vase with foot, two handles, and neck, which served as pi "	"panathenaic prize amphora"@en	"preferred"
Vessel_for_storing_and_transport_liquid_water_with_neck	"Vase with neck and three handles, two for carrying and or "	"hydria"@en	"preferred"
Vessel_for_storing_and_transport_with_neck_with_handle	"Vase with neck and two handles for storing and transport "	"amphora"@en	"preferred"
Vessel_for_storing_and_transport_with_neck_with_handle	"Amphora with foot, with a clearly defined neck."@en	"neck amphora"@en	"preferred"
Vessel_for_storing_and_transport_with_neck_with_handle	"Amphora with foot and a sagging belly."@en	"pelike"@en	"preferred"
Vessel_for_storing_and_transport_with_neck_with_handle	"Belly amphora with round handles and round lip."@en	"type C amphora"@en	"preferred"
Vessel_for_storing_and_transport_with_neck_with_handle	"Belly amphora with flat handles and straight lip."@en	"type A amphora"@en	"preferred"

CQ7: List of English terms with their status and definition using only OTV vocabulary?

Protégé

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
```

```
SELECT ?termName ?status ?def
```

```
WHERE {
    ?subject rdf:type otv:Concept.
    ?subject otv:denotedByTerm ?term.
    ?term otv:termName ?termName.
    ?term otv:termStatus ?status.
    ?term otv:termDefinition ?def.
    ?term otv:language ?lg.
    FILTER (?lg="en")
}
```

```
ORDER BY ?termName
```

termName	status	def
"amphora"	"preferred"	"Vase with neck and two handles for storing and transport liquids and solids."
"bell krater"	"preferred"	"Krater without neck, with upward curling handles placed high on the body."
"belly amphora"	"preferred"	"Amphora with foot, without a clearly defined neck."
"calyx krater"	"preferred"	"Krater without neck, with upward curling handles placed low on the body."
"column krater"	"preferred"	"Krater with a clearly defined neck and column-like handles."
"hydria"	"preferred"	"Vase with neck and three handles, two for carrying and one for pouring, for storing and transport water."
"krater"	"preferred"	"Large open container with a foot and handles, used in ancient Greece for mixing wine with water."
"lekythos"	"preferred"	"Vase for perfume for oil, with narrow neck, with one handle."
"neck amphora"	"preferred"	"Amphora with foot, with a clearly defined neck."
"panathenaic amphora"	"alternative"	"Vase with foot, two handles, and neck, which served as prize in the Panathenaic games. Its names com
"panathenaic prize amphora"	"preferred"	"Vase with foot, two handles, and neck, which served as prize in the Panathenaic games. Its names com
"pelike"	"preferred"	"Amphora with foot and a sagging belly."
"type A amphora"	"preferred"	"Belly amphora with flat handles and straight lip."
"type B amphora"	"preferred"	"Belly amphora with round handles and straight lip."
"type C amphora"	"preferred"	"Belly amphora with round handles and round lip."
"vase"	"preferred"	"Ceramic vessel."
"volute krater"	"preferred"	"Krater with a clearly defined neck and volute-like handles."

CQ8: What is (are) the difference(s) between a 'bell krater' and a 'type B amphora'?

i.e. What are the characteristics of bell kraters which are not characteristics of type B amphorae?

Protégé

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
```

```
SELECT distinct ?diff1Name
```

```
WHERE {?cpt1 rdfs:label "column krater"@en.
      ?cpt1 otv:difference ?diff1.
      ?diff1 rdfs:label ?diff1Name.
      ?cpt2 rdfs:label "type B amphora"@en.
      ?cpt2 otv:difference ?diff2.
      FILTER NOT EXISTS {?cpt2 otv:difference ?diff1}
}
```

```
ORDER BY ?diff1Name
```

"for mixing wine with water"
"with a clearly defined neck"
"with column-like handles"
"with open mouth"

CQ9: What is (are) the difference(s) between a 'type B amphora' and a 'bell krater'?

Protégé

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
```

```
SELECT distinct ?diff2Name
```

```
WHERE {?cpt1 rdfs:label "column krater"@en.
      ?cpt1 otv:difference ?diff1.
      ?cpt2 rdfs:label "type B amphora"@en.
      ?cpt2 otv:difference ?diff2.
      ?diff2 rdfs:label ?diff2Name.
      FILTER NOT EXISTS {?cpt1 otv:difference ?diff2}
}
```

```
ORDER BY ?diff2Name
```

"for storing and transport"
"liquid or solid"
"with round handles"
"with standard belly"
"with straight lip"
"with two handles"
"without a clearly defined neck"

CQ10: What are all the differences between a 'bell krater' and a 'type B amphora'?

Protégé

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

PREFIX otv: <http://www.ontologia.fr/OTB/otv#>

SELECT distinct ?diffName

```
WHERE {
  {?cpt1 rdfs:label "column krater"@en.
    ?cpt1 otv:difference ?diff1.
    ?diff1 rdfs:label ?diff1Name.
    ?cpt2 rdfs:label "type B amphora"@en.
    ?cpt2 otv:difference ?diff2.
    ?diff2 rdfs:label ?diff2Name.
    FILTER NOT EXISTS {?cpt2 otv:difference ?diff1}
    BIND (?diff1Name AS ?diffName) }
  UNION
  { ?cpt1 rdfs:label "column krater"@en.
    ?cpt1 otv:difference ?diff1.
    ?diff1 rdfs:label ?diff1Name.
    ?cpt2 rdfs:label "type B amphora"@en.
    ?cpt2 otv:difference ?diff2.
    ?diff2 rdfs:label ?diff2Name.
    FILTER NOT EXISTS {?cpt1 otv:difference ?diff2}.
    BIND (?diff2Name AS ?diffName)}
}
```

ORDER BY ?diffName

"for mixing wine with water"
"for storing and transport"
"liquid or solid"
"with a clearly defined neck"
"with column-like handles"
"with open mouth"
"with round handles"
"with standard belly"
"with straight lip"
"with two handles"
"without a clearly defined neck"

CQ11: What are the common characteristics of bell kraters and type B amphorae?

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

PREFIX otv: <http://www.ontologia.fr/OTB/otv#>

SELECT distinct ?diff1Name

```
WHERE {
  ?cpt1 rdfs:label "column krater"@en.
  ?cpt1 otv:difference ?diff1.
  ?diff1 rdfs:label ?diff1Name.
  ?cpt2 rdfs:label "type B amphora"@en.
  ?cpt2 otv:difference ?diff1
}
```

ORDER BY ?diff1Name

"vessel"
"with foot"
"with handles"
"with neck"

CQ12: List of the first 10 black-figure vases with their images?

<http://demo.openlinksw.com/sparql>

#List of the first 10 black-figure vases and their images whose reference is greater than 6000

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
PREFIX krat: <http://www.ontologia.fr/OTB/krater#>

SELECT ?label ?img

FROM <http://www.ontologia.fr/OTB/krater.rdf>

WHERE {
  ?object rdf:type otv:Object.
  ?object krat:technique "RED-FIGURE"^^<http://www.w3.org/2001/XMLSchema#string>.
  ?object skos:prefLabel ?label.
  ?object foaf:depiction ?img.
  ?object krat:vase_number_in_Beazley ?num.
  FILTER (?num > 6000)
}

ORDER BY ?label
LIMIT 10
```

label	img
"Beazley-214878"	http://ontologia.fr/Images/GreekVase/Beazley-214878.png
"Beazley-215332"	http://ontologia.fr/Images/GreekVase/Beazleyr-215332.png
"Beazley-215332"	http://ontologia.fr/Images/GreekVase/Beazley-215332.png
"Beazley-215424"	http://ontologia.fr/Images/GreekVase/Beazley-215424.png
"Beazley-217564"	http://ontologia.fr/Images/GreekVase/Beazleyr-217564.png
"Beazley-217564"	http://ontologia.fr/Images/GreekVase/Beazley-217564.png
"Beazley-218065"	http://ontologia.fr/Images/GreekVase/Beazleyr-218065.png
"Beazley-218065"	http://ontologia.fr/Images/GreekVase/Beazley-218065.png

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
PREFIX krat: <http://www.ontologia.fr/OTB/krater#>
```

label	
"Beazley-214878"	Beazley-214878.png
"Beazley-215332"	Beazleyr-215332.png
"Beazley-215332"	Beazley-215332.png
"Beazley-215424"	Beazley-215424.png
"Beazley-217564"	Beazley-217564.png
"Beazley-217564"	Beazleyr-217564.png
"Beazley-218065"	Beazley-218065.png
"Beazley-218065"	Beazleyr-218065.png

```
SELECT ?label ?img
```

```
WHERE {
  ?object rdf:type otv:Object.
  ?object krat:technique "RED-FIGURE"^^<http://www.w3.org/2001/XMLSchema#string>.
  ?object skos:prefLabel ?label.
  ?object foaf:depiction ?img.
  ?object krat:vase_number_in_Beazley ?num.
  FILTER (?num > 6000)
}
ORDER BY ?label
LIMIT 10
```


```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
PREFIX krat: <http://www.ontologia.fr/OTB/krater#>
```

label	
"Beazley-214878"	Beazley-214878.png
"Beazley-215332"	Beazleyr-215332.png
"Beazley-215332"	Beazley-215332.png
"Beazley-215424"	Beazley-215424.png
"Beazley-217564"	Beazley-217564.png
"Beazley-217564"	Beazleyr-217564.png
"Beazley-218065"	Beazley-218065.png
"Beazley-218065"	Beazleyr-218065.png

```
SELECT ?label ?img
```

```
WHERE {
  ?object rdf:type otv:Object.
  ?object krat:technique ?tech.
  ?object skos:prefLabel ?label.
  ?object foaf:depiction ?img.
  ?object krat:vase_number_in_Beazley ?num.
  FILTER (?tech="RED-FIGURE")
  FILTER (?num > 6000)
}
ORDER BY ?label
LIMIT 10
```


5. Export OntoLex-Lemon



<http://demo.openlinksw.com/sparql>

1) Return the list of English terms with their definition using OntoLex-Lemon:

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

PREFIX ontolex: <http://www.w3.org/ns/lemon/ontolex#>

SELECT ?term ?definition

FROM <http://www.ontologia.fr/OTB/krater.rdf>

WHERE {
 ?lexEntry rdf:type ontolex:LexicalEntry.
 ?lexEntry ontolex:lexicalForm ?lexForm.
 ?lexForm ontolex:writtenRep ?term.
 ?lexEntry ontolex:evokes ?lexCpt.
 ?lexCpt skos:definition ?definition.
 FILTER (lang(?term) = 'en')
 FILTER (lang(?definition) = 'en')
}

ORDER BY ?term

LIMIT 10

How to Query in SPARQL an Ontoterminology

term	definition
"amphora"@en	"Vase with neck and two handles for storing and transport liquids and solids."@en
"bell krater"@en	"Krater without neck, with upward curling handles placed high on the body."@en
"belly amphora"@en	"Amphora with foot, without a clearly defined neck."@en
"calyx krater"@en	"Krater without neck, with upward curling handles placed low on the body."@en
"column krater"@en	"Krater with a clearly defined neck and column-like handles."@en
"hydria"@en	"Vase with neck and three handles, two for carrying and one for pouring, for storing a
"krater"@en	"Large open container with a foot and handles, used in ancient Greece for mixing wine
"lekythos"@en	"Vase with neck for perfume."@en
"neck amphora"@en	"Amphora with foot, with a clearly defined neck."@en
"panathenaic amphora"@en	"Vase with foot, two handles, and neck, which served as prize in the Panathenaic games